

- Bureau d'étude du cours d'hydrologie littorale et côtière -

Réalisation d'un modèle de transformation
de la houle 1D pour l'étude d'un projet de
rechargement de plage



Table des matières

- Bureau d'étude du cours d'hydrologie littorale et côtière -	1
Introduction.....	3
I. Théorie et Programmation.....	3
I.1. Rappel des hypothèses.....	3
I.2. Théorie.....	3
I.3. Algorithme.....	4
II. Validation des résultats.....	10
III. Application à un cas réel : La plage de Sète.....	13
Conclusion.....	18

Introduction

Nous effectuons ce bureau d'étude dans le cadre du cours « Hydraulique Littoral et Côtier » de la troisième année Hydraulique option Sciences de l'Eau et de l'Environnement de l'ENSEEIH. L'objectif de ce BE est de créer un modèle de transformation de la houle le long d'un profil 1D caractérisé par sa profondeur $d(x)$ en vue de l'étude d'un projet réel de rechargement en sable de la plage de Sète.

Notre modèle devra calculer la hauteur de la houle $H(x)$ en tout point du profil, connaissant la hauteur de houle H_0 et la période T à la frontière. Dans une première partie, nous rappellerons succinctement les équations régissant le phénomène de la houle et du déferlement puis nous présenterons le programme que nous avons sous Fortran. Dans une seconde partie, nous vérifierons la validité de notre modèle en comparant nos résultats avec les mesures expérimentales réalisées par C. Dulou dans un petit canal à houle. Enfin dans une dernière partie, nous chercherons à l'aide de notre modélisation, nous regarderons quelles sont les meilleures conditions pour le rechargement en sable.

I. Théorie et Programmation

1.1. Rappel des hypothèses

Nous voulons modéliser la transformation de la houle le long d'un profil 1D. Nous allons pour cela séparer deux zones bien distinctes dans notre modélisation, définies par le critère de déferlement. La première zone est la zone de houle linéaire, la seconde, située au delà du point critique de déferlement est la zone de déferlement. Pour pouvoir modéliser ces phénomènes, nous avons besoin d'un certain nombre d'hypothèses.

- Le problème est stationnaire $\rightarrow H(x,t) = H(x)$.
 - pas de viscosité.
 - la masse volumique de l'eau, considérée comme incompressible, est constante,
 - pente du fond faible \rightarrow solution sur fond plat valable en première approximation, pas de réflexion de la houle.
 - conservation de la période de la houle le long du profil.
 - cambrure de la houle faible \rightarrow Modèle linéaire valable
- Physiquement, la période est de l'ordre de 10 secondes.

1.2. Théorie

Avant le déferlement

La conservation du flux d'énergie $F(x)$ s'écrit $\frac{\partial F}{\partial x} = 0$ avec $F(x) = E(x)C_g(x)$ où

$E(x) = \frac{1}{8} \rho_0 g H(x)^2$ représente la densité d'énergie par unité de longueur dans le cadre de la théorie linéaire (ρ_0 est la densité du fluide et g l'accélération de la pesanteur), et C_g la vitesse de groupe de la houle, $C_g(x) = \frac{1}{2} \frac{\omega}{k}(x) \left(1 + \frac{2k(x)d(x)}{sh(2k(x)d(x))}\right)$.

Pour le calcul de k , nous avons utilisé la forme approchée de la relation de dispersion dans le cas d'un fond plat. On admet que cette relation est valable, dans notre étude, en considérant que le fond, à pente de variation faible, est localement plat. Nous utiliserons:

$$(k(x) * d(x))^2 = y^2 + \frac{y}{1 + 0,666 y + 0,355 y^2 + 0,160 y^3 + 0,0632 y^4 + 0,0217 y^5 + 0,00654 y^6}$$

avec $y = \frac{\omega d(x)}{g}$

On peut maintenant calculer $H(x)$ le long du profil 1D et vérifier si elle provoque le déferlement.

Critère de déferlement

On utilise deux critères pour définir si il y a déferlement ou non: le critère de Miche et le critère de Weggel.

Critère de Miche :

$$\left(\frac{H(x)}{\lambda}\right)_{max} = 0,14 th(k(x)d(x))$$

pour $\left(\frac{H(x)}{\lambda}\right)_{max} \leq 0,14 th(k(x)d(x))$, on est dans la zone de houle linéaire.

Critère de Weggel :

$$\left(H \frac{d(x)}{d}\right)_{max} = b(m) - a(m) \frac{H(x)}{gT^2}$$

où m est la pente.

Le déferlement:

Pour calculer la hauteur de houle dans la zone de déferlement, on utilise la relation liant la hauteur de houle avec la profondeur ainsi que la paramétrisation de la dissipation proposée par Le Méhauté.

$$\frac{\partial F}{\partial x} = \frac{-\beta}{4} \rho_0 g f \frac{H(x)^3}{d(x)}$$

où f est la fréquence de la houle ($1/T$) et β est un coefficient que nous prendrons égal à 1.

1.3. Algorithmme

On programme sous Fortran.

```

!*****
!          PROGRAMME PRINCIPAL
!*****

! Calcul de H en fonction du fond et des variables présentes dans
    
```

```

! le fichier annexe "donnees"

! Déclaration des variables
  IMPLICIT NONE
  INTEGER n,d_OPEN,r_OPEN,i,j,Miche,critere,choix_critere,Weggel
  REAL H0,T,x,d,calcul_k,k,profondeur
  REAL H,xd,H_avant,H_apres,alpha
  ALLOCATABLE x(:)
  ALLOCATABLE d(:)
  ALLOCATABLE H(:)
  d_OPEN=1
  r_OPEN=2

! Lecture du fichier "donnees" contenant les conditions aux limites (H0
! et T) ainsi que l'évolution de la bathymétrie en fonction de x
  OPEN(d_OPEN,file='donnees')
  READ(d_OPEN,*) H0
  READ(d_OPEN,*) T
  READ(d_OPEN,*) alpha ! Coefficient de la Loi de Le Méhauté
  READ(d_OPEN,*) n
  ALLOCATE (x(n),d(n),H(n))
  DO i=1,n
    READ(d_OPEN,*) x(i),d(i)
    d(i)=-d(i)
  ENDDO

! Choix de la methode pour la calculer le critere de deferlement :
! 0 -> MICHE || 1 -> WEGGEL
  choix_critere = 1

! Calcul de l'amplitude H(x) avant déferlement par la théorie de houle
! linéaire
! Initialisation
  i=1
  H(1)=H0
  k=calcul_k(d(1),T)

  IF(choix_critere==0) THEN
    ! Calcul de H tant que le critère de Miche est vérifié
    critere=Miche(k,d(1),H0)
    DO WHILE(critere==1)
      i=i+1
      H(i)=H_avant(H0,d(1),T,d(i))
      k=calcul_k(d(i),T)
      critere=Miche(k,d(i),H(i))
    ENDDO
  
```

```

ELSEIF(choix_critere==1) THEN
  ! Calcul de H tant que le critère de Weggel est vérifié
  critere=Weggel(k,x(1),x(2),d(1),d(2),H0,T)
  DO WHILE(critere==1)
    i=i+1
    H(i)=H_avant(H0,d(1),T,d(i))
    k=calcul_k(d(i),T)
    critere=Weggel(k,x(i),x(i+1),d(i),d(i+1),H(i),T)
  ENDDO
ENDIF

! Mise en mémoire de la valeur de l'abscisse de déferlement
xd=x(i)
WRITE(*,*) 'xdeferlement=',xd

! Calcul de l'amplitude H(x) après déferlement par introduction d'un
! terme de dissipation, pour x>xd
DO j=i,n
  H(j)=H_apres(x(j),x(j-1),H(j-1),d(j),d(j-1),T,alpha)
ENDDO

! Stockage des résultats dans le fichier nommé "resultats"
OPEN(r_OPEN,file='resultats')
WRITE(r_OPEN,*) n
DO i=1,n
  WRITE(r_OPEN,*) x(i),H(i)
ENDDO
DEALLOCATE (x,d,H)
END

!*****
!      FONCTION  calcul_k
!*****

! Calcul de k(x) par une formule approchée

FUNCTION calcul_k(profondeur,periode)
  REAL profondeur,periode,calcul_k,pulsation,somme
  REAL d1,d2,d3,d4,d5,d6,y
  REAL e,k1,k2
  pulsation=2*3.14159/periode
  y=pulsation*pulsation*profondeur/9.81
  d1=0.666*y
  d2=0.355*y*y

```

```

d3=0.160*y*y*y
d4=0.0632*y*y*y*y
d5=0.0217*y*y*y*y*y
d6=0.00654*y*y*y*y*y*y
somme=d1+d2+d3+d4+d5+d6
calcul_k=(y*y+y/(1+somme))/(profondeur*profondeur)
calcul_k=sqrt(calcul_k)

```

```

! Calcul de k(x) par la méthode de Newton à partir de la formule
! de dispersion

```

```

e=100
k1=calcul_k
DO WHILE (e>0.1)
k2=k1-disp(k1,profondeur,periode)/d_disp(k1,profondeur,periode)
e=abs(k2-k1)
k1=k2
ENDDO
calcul_k=k2
END

```

```

!*****

```

```

!      FONCTION dispersion

```

```

!*****

```

```

FUNCTION disp(k,h,T)
  REAL k,h,T,disp,w
  w=2*3.14159/T
  disp=9.81*k*tanh(k*h)-(w*w)
END

```

```

!*****

```

```

!      FONCTION dérivée de la relation de dispersion

```

```

!*****

```

```

FUNCTION d_disp(k,h,T)
  REAL k,h,T,d_disp,w,num,deno
  w=2*3.14159/T
  num=(exp(4*h*k)+4*h*k*exp(2*h*k)-1)
  deno=(exp(2*h*k)+1)*(exp(2*h*k)+1)
  d_disp=9.81*num/deno
END

```

```

!*****

```

```

!      FONCTION calcul_cg

```

```

!*****

```

```
! Calcul de la vitesse de groupe cg(x) à partir de la relation de
! dispersion, connaissant k(x)
```

```
FUNCTION calcul_cg(profondeur,periode)
  REAL pulsation,k,calcul_cg,denom,periode
  pulsation=2*3.14159/periode
  k=calcul_k(profondeur,periode)
  denom=sinh(2.0*k*profondeur)
  calcul_cg=0.5*(pulsation/k)*(1+(2*k*profondeur)/(denom))
END
```

```
!*****
```

```
!      FONCTION  H_avant
```

```
!*****
```

```
! Calcul de l'amplitude de la vague H(x) avant déferlement connaissant
! cg(x)
```

```
FUNCTION H_avant(H0,d0,T,d)
  REAL d,cg,H_avant
  REAL cg0,H0,d0,T
  cg0=calcul_cg(d0,T)
  cg=calcul_cg(d,T)
  H_avant=sqrt(cg0/cg*H0*H0)
END
```

```
!*****
```

```
!      FONCTION  Miche
```

```
!*****
```

```
! Estime si l'amplitude H(x) calculée vérifie ou non le critère de
! Miche, pour savoir si la vague a déferlé ou non
```

```
FUNCTION Miche(k,profondeur,H)
  REAL k,profondeur,H,L,lim
  INTEGER Miche

  L=2*3.14159/k
  lim=0.14*tanh(k*profondeur)

  WRITE(*,*) 'k,H/L, lim',k,H/L,lim

  IF (H/L<lim) THEN
    Miche=1
  ELSE
```



```

        Miche=0
    ENDDIF

END

!*****
!    FONCTION  Weggel
!*****

! Estime si l'amplitude H(x) calculée vérifie ou non le critère de
! Weggel, pour savoir si la vague a déferlé ou non

FUNCTION Weggel (k,x1,x2,profondeur,d2,H,periode)
REAL k,x1,x2,profondeur,d2,H,periode,penete,A,B,lim
INTEGER Weggel

penete=(d2-profondeur)/(x2-x1)
A=43.8*(1-exp(-19*penete))
B=1.56/(1+exp(-19.5*penete))
lim=B-A*H/(9.81*periode*periode)

WRITE(*,*) 'k,H/profondeur,lim',k,H/profondeur,lim

IF (H/profondeur<lim) THEN
    Weggel=1
ELSE
    Weggel=0
ENDIF

END

!*****
!    FONCTION  H_après
!*****

! Calcul de l'amplitude de la vague H(x) après déferlement connaissant
! cg(x) par l'introduction d'un terme de dissipation

FUNCTION H_apres(x2,x1,H1,d2,d1,T,alpha)
REAL x1,x2,H1,d1,d2,H_apres,alpha,T,D
REAL cg1,cg2

cg1=calcul_cg(d1,T)
cg2=calcul_cg(d2,T)

! Formule de Le Mehauté
D=(alpha/4.0)*(9810.0/T) *(H1*H1*H1)/d1


```

```
H_apres=sqrt((cg1*H1*H1-(8*(x2-x1)*D/9810.0))/cg2)
END
```

II. Validation des résultats


On utilise les mesures expérimentales réalisées par C. Dulou dans un petit canal à houle pour valider le modèle. Nous avons à notre disposition 3 séries de données expérimentales pour des houles différentes, caractérisées par leurs hauteurs H_0 et leurs périodes.

En particulier, on veut vérifier la validité des critères de déferlement par leurs aptitudes à prédire la position du point de déferlement pour les différentes hauteurs de houles considérées. Pour cela, nous allons présenter les données expérimentales aux simulations faites avec le critère de Miche et celui de Weggel. On convertit toutes les données en unités S.I.. Commençons par présenter la bathymétrie du fond.



Cas expérimental 1, hauteur de houle 8,2 mm:

En regardant le graphique de la houle en fonction du temps, on identifie une période de 0,67s. En comparant les hauteurs de houle expérimentales et simulées, on obtient:




Abscisse de déferlement:

expérimentalement: 2,338 m
 Miche: 2,403 m
 Erreur de Miche: 2,78%
 Weggel: 2,273m
 Erreur Weggel: 2,78%

Cas expérimental 2, hauteur de houle 11,4 mm:


En regardant le graphique de la houle e fonction du temps, on identifie une période de 0,70s.



Abscisse de déferlement:
 expérimentalement: 2,113 m
 Miche: 2,176 m
 Erreur de Miche: 2,98%
 Weggel: 2,109 m
 Erreur de Weggel: 0,1%

Cas expérimental 3, hauteur de houle 13,5 mm:

En regardant le graphique de la houle e fonction du temps, on identifie une période de 0,71s.



Abscisse de déferlement:

expérimentalement: 2,006 m

Miche: 2,157 mm

Erreur de Miche: 7,53%

Weggel: 1,996

Erreur de Weggel: 0,4%


On note que la modélisation avec le critère de Miche donne de bons résultats quand la houle n'est pas trop haute. Plus la houle augmente, plus le critère de Miche donne un résultat éronné pour le point de déferlement. A l'opposé, le critère de Weggel donne de meilleurs résultats quand la houle devient forte et des résultats identiques à celui du critère de Miche pour des houles moyennement forte. Nous choisissons donc pour la suite, le critère de Weggel pour nos simulations. On remarque tout de même que les deux critères donnent une bonne estimation générale du phénomène.

III. Application à un cas réel : La plage de Sète

La plage de Sète, comme la plupart des cotes sablonneuses du littoral méditerranéen est en péril. Sa dégradation est due à un déficit d'apport en sédiment. La plage est régulièrement soumise aux assauts de la houle et des tempêtes. Pour cette partie, on utilise le critère de déferlement de Weggel.


On note la présence de barres dans la bathymétrie naturelle: une très importante à 575 m et une plus petite à 690m. On envisage deux solutions pour changer cette bathymétrie et atténuer les effets de la houle sur la plage en la rendant moins forte. On évitera ainsi une érosion trop intense. Les deux propositions sont des rechargements de ces barres:

- Scénario A: La première barre (externe) est rechargée d'un mètre.
- Scénario B: La première barre est rechargée d'un mètre et la seconde de 50 cm.




Tracé de la bathymétrie avant la déformation du fond

On veut maintenant simuler les effets de ces scénarios sur différents types de houle: La houle de tempête ($H_0=2,5m$, $T=7$) et la houle de tombant de tempête ($H_0=1m$, $T=6,5s$). Nous cherchons quel scénario minimise la hauteur de houle entre la barre interne et la plage lors de la houle de tempête pour limiter le transport de sable par le courant de retour. Et nous voulons maximiser cette même hauteur lors de la houle de tombant de tempête afin de maximiser l'apport de sable.




Impact des deux houles sur la bathymétrie




Prévisions de l'impact de deux types de houle sur le scénario d'aménagement A

On peut constater que le changement de comportement de la houle avec le scénario A est faible par rapport à l'état naturel.




Prévisions de l'impact de deux types de houle sur le scénario d'aménagement B

On remarque un changement beaucoup plus flagrant du comportement de la houle avec le scénario B.



Graphique de l'impact de chaque fond sur une houle de tempête.

On remarque que le scénario B déplace le point de déferlement bien avant les barres : la hauteur d'eau est donc largement minimisée entre la barre interne et la plage. Le scénario B est beaucoup plus efficace que le scénario A pour remplir l'objectif de minimisation de la perte de sable suite à une tempête.



Graphique de l'impact de chaque fond sur une houle de tombant de tempête.

On note que le scénario B minimise aussi la hauteur de houle lors du tombant de tempête ce qui est à l'opposé de nos souhaits.

Conclusion

Après quelques recherches sur Internet(Le Lido, De grands enjeux un grand projet, Thau Agglomeration, dossier presse 27 juin 2007), nous constatons que la majorité des aménagements mis en place pour protéger les littoraux de l'érosion ont pour principal but de diminuer la houle. Bien que le scénario A permettent un meilleur apport en sable, le scénario B est globalement plus efficace: il diminue la houle de façon plus importante en tempête par rapport à la façon dont le scénario A augmente la houle en tombant de tempête. Pour cette raison, nous proposons le scénario B et si possible un réchargement en sable de la plage pour compenser les effets de transport de sable par le courant retour, certe minimisé mais pas entièrement annulé.